Using the "B" Switch to Add Special Unprintable Characters to a QR Bar Code
Summarizing the B switch:
Bxxxxdata
xxxx refers to the number of characters represented
The example below shows B0002 (two characters) and B0001 (one character).
^XA
^FO100,90^A0N,30,30^FDFIRST BARCODE NULL STX 12345 NULL 67890^FS
^FO100,150
^BQN,2,5
^FO120,270^FH_^FDMM,B0002_00_02,N12345,B0001_00,N67890^FS
^XZ
^FH_ is the Field Hex command. A ZPL printer will treat the two digits following _ as Hex
values. The Hex value for _00 is NULL and the Hex value for _02 is STX.
For the "B" switch, _## specifying a Hex value counts as just one character.
Also, in this example using the N for Numeric means the data characters of 12345 and 67890 do
not have to be counted in the number of characters specified for B.
Scanner results are:
[NULL] [STX]12345 [NULL]67890 [ CR] [ LF]

This example shows B0013 (thirteen characters)
^XA
^FO100,90^A0N,30,30^FDSECOND BARCODE NULL STX 12345 NULL 67890^FS
^BQN,2,5
^FO420,420^ FH _^FDMM,B0013_00_0212345_0067890^FS
Since there is no N, the _##'s each counting as one character and the 10 digits (12345 & 67890)
combined count as 13 characters
Scanner results are the same:
[NULL] [STX]12345 [NULL]67890 [ CR] [ LF]

This example shows B0021 (twenty one characters)
^FO100,50^A0N,30,30^FD21CHARACTERS IN THIRD BARCODE^FS
^BQN,2,5
^FO120,620^ FH _^FDMM,B0021123456789012345678901^FS
^XZ
Since there are no _##'s this example uses B0021 just to account for the 21 digits. If you
miscounted and specified 18, for example, the last 3 digits would truncate from the bar code that
prints and scans.

Here are examples for adding special characters that will perform Carriage Return line
feeds when you scan and print the bar codes.
In this first example an A precedes each set of Alpha Numeric characters. Similar to the N in the
previous example, this means that every character does not have to be counted. The B switch
values specify 0002 as the number of characters for each of the Hex 0d (carriage return) and the
Hex 0a (line feed).
^XA
^MMT
^PW609
^LL0406
^LS0
^FT229,324^BQN,2,4
^FH_^FDMM,Ahttp://www.commscope.com,B0002_0d_0a,ATypeID:02,B0002_0d_0a,AMater
ialDesc:ION-U L 7-8-85-17p-Vac,B0002_0d_0a^FS
^PQ1,0,1,Y^XZ
Scanner results are:
http://www.commscope.com [ CR][ LF]
TypeID:02 [ CR][ LF]
MaterialDesc:ION-U L 7-8-85-17p-Vac [ CR][ LF]
[ CR][ LF]

In this example with no A preceding the Alpha Characters all 23 letters AND SPACES between
them are specified in the "B" Switch value of B0023
^XA
^FT100,60^BQN,2,10^FH^FDMM,B0023first WORD_0d_0aSECOND WORD^FS
^XZ
Scanner results are:
first WORD [ CR][ LF]
SECOND WORD [ CR][ LF]
[ CR][ LF]

NOTE - Using ^FT and not ^FO will print the QR Code in the expected position on a printed label

